
CLASS 1-3 CLASS 3	CHAPTERS 1-2, ASSIGNMENT #1	1/22 : 1/29 1/29
CLASS 4	QUIZ 1	2/2
CLASS 4-7 CLASS 7	CHAPTER 3 ASSIGNMENT #2	2/2 : 2/12 2/12
CLASS 8	QUIZ 2 (ONLY QUIZ, NO CLASS)	2/23
CLASS 9-11 CLASS 11	CHAPTERS 4-5 ASSIGNMENT #3	2/26 : 3/5 3/5
CLASS 12	QUIZ 3	3/9
CLASS 12-14 CLASS 14	CHAPTER 6 (PRINT, FREQ, MEANS) ASSIGNMENT #4	3/9 : 3/16 3/16
CLASS 15	QUIZ 4	3/19
CLASS 15-16 CLASS 16	PART OF CHAPTER 7 (TO PAGE 136) ASSIGNMENT #5	3/19 : 3/23 3/23
CLASS 17	QUIZ 5	3/26
CLASS 17-19 CLASS 19	CHAPTERS 7-8 ASSIGNMENT #6	3/26 : 4/2 4/2
CLASS 20	QUIZ 6	4/6
CLASS 20-22 CLASS 22	CHAPTERS 9-10 ASSIGNMENTS #7, #8	4/6 : 4/20 4/20
CLASS 23	QUIZ 7	4/23
CLASS 23-25 CLASS 25	CHAPTER 11, "LOOSE ENDS" ASSIGNMENT #9	4/23 : 4/30 4/30
CLASS 26	QUIZ 8	5/4
OPTIONAL FINAL	REGULAR CLASS TIME	5/11

GRADE BASED ON QUIZZES**IF YOU TAKE THE FINAL, THROW OUT ONE QUIZ****30 (OR 35)% OF GRADE BASED ON FINAL + 70 (OR 65)% OF GRADE BASED ON QUIZZES ... WHATEVER COMBINATION GIVES THE HIGHEST GRADE**

EXTRA STUFF

REPLACE NEW PAGE WITH A SINGLE CHARACTER (YOU'LL GET A ROW OF THESE CHARACTERS RATHER THAN A NEW PAGE -> SAVES PAPER)

```
options formdlm = '-';
```

LEFT-JUSTIFY SAS OUTPUT

```
options nocenter;
```

TO RESET

```
options formdlm = '';
```

that's 2 consecutive quotes
with nothing in between

```
options center;
```

TO GET RID OF ALL THE TITLE STATEMENTS YOU MAY HAVE ENTERED

```
title;
```

TO GET RID OF SAS-SUPPLIED EXTRA TITLES SUCH AS ...

THE MEANS PROCEDURE

```
ods noproctitle;
```

FIRSTOBS AND OBS GLOBAL STATEMENT OPTIONS

in addition to the FIRSTOBS and OBS INFILE and data set options covered in the notes, you can also use FIRSTOBS and OBS in a GLOBAL OPTIONS statement ... the examples use a SAS-supplied data set that has 19 observations

remember that the default values are ... FIRSTOBS=1, OBS=MAX

```
title 'DATA SET SASHELP.CLASS';
proc print data=sashelp.class;
run;

* use a GLOBAL OPTIONS statement and set FIRSTOBS and OBS values;

options firstobs=11 obs=15;

title 'OPTIONS STATEMENT USED - FIRSTOBS=11 OBS=15';
proc print data=sashelp.class;
run;

proc means data=sashelp.class;
run;

* override the GLOBAL OPTION with a DATA SET OPTION;

title 'OPTIONS STATEMENT USED - FIRSTOBS=11 OBS=15 AND OBS=MAX DATA SET OPTION';
proc print data=sashelp.class (obs=max);
run;

proc means data=sashelp.class (obs=max);
run;

* reset the GLOBAL OPTION for OBS to MAX;

options obs=max;

title 'OPTIONS STATEMENT USED - RESET LAST OBSERVATION WITH OBS=MAX';
proc print data=sashelp.class;
run;

* reset the GLOBAL OPTION for FIRSTOBS to 1;

options firstobs=1;

title 'OPTIONS STATEMENT USED - RESET FIRST OBSERVATION WITH FIRSTOBS=1';
proc print data=sashelp.class;
run;

proc means data=sashelp.class;
run;
```

IF VERSUS WHERE WITH A NEW VARIABLE IN A DATA STEP

these examples should be added to the discussion of WHERE versus IF at the end of chapter 4 in the notes ... given a data set with gestation in days, create a new data set with only those observations less than 37 weeks gestation (less than 259 days)

```
* some test data - gestaTion in days;

data test;
input ges @@;
datalines;
140 196 208 210 211 245 246 280 281 253 257 258 259
;
run;

title 'ORIGINAL DATA';
proc print data=test;
run;
```

ORIGINAL DATA

Obs	ges
1	140
2	196
3	208
4	210
5	211
6	245
7	246
8	280
9	281
10	253
11	257
12	258
13	259

```
data new;
set test;
gwk = ges/7;
if gwk lt 37;
format gwk 3.;
run;

title1 'GESTATION LESS THAN 37 WEEKS';
title2 'IF GWK LT 37; (ROUND GWK WITH FORMAT 3.)';
proc print data=new;
run;
```

GESTATION LESS THAN 37 WEEKS IF GWK LT 37; (ROUND GWK WITH FORMAT 3.)

Obs	ges	gwk
1	140	20
2	196	28
3	208	30
4	210	30
5	211	30
6	245	35
7	246	35
8	253	36
9	257	37
10	258	37

```
*
use the INT function to select
only the integer portion of GWK
drop the decimal places, do not round
;
```

```
data new;
set test;
gwk = int(ges/7);
if gwk lt 37;
run;
```

GESTATION LESS THAN 37 WEEKS IF GWK LT 37; (TRUNCATE GWK WITH INT FUNCTION)

Obs	ges	gwk
1	140	20
2	196	28
3	208	29
4	210	30
5	211	30
6	245	35
7	246	35
8	253	36
9	257	36
10	258	36

```
title1 'GESTATION LESS THAN 37 WEEKS';
title2 'IF GWK LT 37; (TRUNCATE GWK WITH INT FUNCTION)';
proc print data=new;
run;
```

```

*
WHERE statement DOES NOT work since GWK is NOT in the data set TEST
where gwklt 37
;

```

```

data new;
set test;
gwkl = int(ges/7);
where gwkl lt 37;
run;

```

```

760 data new;
761 set test;
762 gwkl = int(ges/7);
763 where gwkl lt 37;
ERROR: Variable gwkl is not on file WORK.TEST.
764 run;

```

```

*
this WHERE statement WORKS since GES is in
the data set TEST
where ges lt 259;
;

```

```

data new;
set test;
gwkl = int(ges/7);
where ges lt 259;
run;

```

```

GESTATION LESS THAN 37 WEEKS
WHERE GES LT 259; (TRUNCATE GWK WITH INT FUNCTION)

```

Obs	ges	gwkl
1	140	20
2	196	28
3	208	29
4	210	30
5	211	30
6	245	35
7	246	35
8	253	36
9	257	36
10	258	36

```

title1 'GESTATION LESS THAN 37 WEEKS';
title2 'WHERE GES LT 259; (TRUNCATE GWK WITH INT FUNCTION)';
proc print data=new;
run;

```

```

*
this WHERE statement WORKS since GES is in
the data set TEST
where ges/7 lt 37;
;

```

```

data new;
set test;
gwkl = int(ges/7);
where ges/7 lt 37;
run;

```

```

GESTATION LESS THAN 37 WEEKS
WHERE GES/7 LT 37; (TRUNCATE GWK WITH INT FUNCTION)

```

Obs	ges	gwkl
1	140	20
2	196	28
3	208	29
4	210	30
5	211	30
6	245	35
7	246	35
8	253	36
9	257	36
10	258	36

```

title1 'GESTATION LESS THAN 37 WEEKS';
title2 'WHERE GES/7 LT 37; (TRUNCATE GWK WITH INT FUNCTION)';
proc print data=new;
run;

```